ТОМСКИЙ ПОЛИТЕХНИЧЕСКИЙ УНИВЕРСИТЕТ

tpu.ru

Углубленный курс информатики

Численные методы решения систем обыкновенных дифференциальных уравнений



Чузлов Вячеслав Алексеевич

к.т.н., доцент ОХИ ИШПР

МЕТОД ЭЙЛЕРА

- Пусть дана система дифференциальных уравнений:

$$\begin{cases} \dfrac{dy_1}{dx} = f_1(x, y_1, y_2) \\ \dfrac{dy_2}{dx} = f_2(x, y_1, y_2) \end{cases}$$

с начальными условиями:

$$y_1|_{x=x_0} = y_{01}$$

$$y_2|_{x=x_0} = y_{02}$$

$$y_1 = y_1(x) - \ ?$$

$$y_2 = y_2(x) - \ ?$$

МЕТОД ЭЙЛЕРА

Формула Эйлера:

$$\begin{cases} y_{(i+1),1} = y_{i,1} + h \cdot f_1(x_i, y_{i,1}, y_{i,2}) \\ y_{(i+1),2} = y_{i,2} + h \cdot f_2(x_i, y_{i,1}, y_{i,2}) \\ x_{i+1} = x_i + h \end{cases}$$

где ***h*** – шаг вычисления;

$f(x_i, y_{i,j})$ – правая часть дифференциального уравнения

ПРИМЕР

Решить систему дифференциальных уравнений первого порядка:

$$\begin{cases} \dfrac{dy_1}{dx} = y_2 \\ \dfrac{dy_2}{dx} = e^{-xy_1} \end{cases}$$

методом Эйлера на отрезке *[0; 1]* с шагом $h = 0.1$.

Начальные условия: $x_0 = 0$; $y_1(0) = 0$; $y_2(0) = 0$.

РЕШЕНИЕ

Система дифференциальных уравнений:

$$\begin{cases} \dfrac{dy_1}{dx} = y_2 \\ \dfrac{dy_2}{dx} = e^{-xy_1} \end{cases}$$

Метод Эйлера:

$$\begin{cases} y_{(i+1),1} = y_{i,1} + h \cdot f_1(x_i, y_{i,1}, y_{i,2}) \\ y_{(i+1),2} = y_{i,2} + h \cdot f_2(x_i, y_{i,1}, y_{i,2}) \\ x_{i+1} = x_i + h \end{cases}$$

i	x	$y_{i,1} = y_{(i-1),1} + h \cdot y_{(i-1),2}$	$y_{i,2} = y_{(i-1),2} + h \cdot e^{-x_{i-1} \cdot y_{(i-1),1}}$
0	0.0	0.0000	0.0000
1	0.1	0.0000	0.1000
2	0.2	0.0100	0.2000
3	0.3	0.0300	0.2998
4	0.4	0.0600	0.3989
5	0.5	0.0999	0.4965
6	0.6	0.1495	0.5917
7	0.7	0.2087	0.6831
8	0.8	0.2770	0.7695
9	0.9	0.3539	0.8496
10	1.0	0.4389	0.9223

РЕАЛИЗАЦИЯ

```
function equations(x: real; y: array of real): array of real;
begin
  result := ArrFill(y.Length, 0.0);
  result[0] := y[1];
  result[1] := exp(-x * y[0])
end;


function Eiler(f: function(x: real; y: array of real): array of real;
               x0, xf: real; y0: array of real;
               h: real): array of array of real;
begin
  var count := Trunc((xf - x0) / h) + 1;
  SetLength(result, count);
  for var i := 0 to result.High do
    SetLength(result[i], y0.Length);

  var x := x0;
  var right_parts := ArrFill(y0.Length, 0.0);
  result[0] := y0;
  for var i := 1 to count-1 do
  begin
    right_parts := f(x, result[i-1]);
    for var j := 0 to result[i].High do
      result[i][j] := result[i-1][j] + h * right_parts[j];
    x += h
  end
end;
```

```
begin
  var y0 := Arr(0.0, 0.0);
  var y := Eiler(equations, 0, 1, y0, 0.1);

  for var i := 0 to y.High do
  begin
    for var j := 0 to y[i].High do
      write(y[i][j]:8:4);
    writeln
  end
end.
```

```
  0.0000  0.0000
  0.0000  0.1000
  0.0100  0.2000
  0.0300  0.2998
  0.0600  0.3989
  0.0999  0.4965
  0.1495  0.5917
  0.2087  0.6831
  0.2770  0.7695
  0.3539  0.8496
  0.4389  0.9223
```

МЕТОД РУНГЕ-КУТТЫ

$$\begin{cases} y_{(i+1),1} = y_{i,1} + (k_{11} + 2 \cdot k_{21} + 2 \cdot k_{31} + k_{41}) \cdot \ h \ / \ 6 \\ y_{(i+1),2} = y_{i,2} + (k_{12} + 2 \cdot k_{22} + 2 \cdot k_{32} + k_{42}) \cdot h \ / \ 6 \\ x_{i+1} = x_i + h \end{cases}$$

$$k_{11} = f_1(x, y_1, y_2)$$

$$k_{21} = f_1(x + h/2\,, y_1 + k_{11} \cdot h/2\,, y_2 + k_{12} \cdot h/2)$$

$$k_{31} = f_1(x + h/2\,, y_1 + k_{21} \cdot h/2\,, y_2 + k_{22} \cdot h/2)$$

$$k_{41} = f_1(x + h, y_1 + k_{31}, y_2 + k_{32})$$

$$k_{12} = f_2(x, y_1, y_2)$$

$$k_{22} = f_2(x + h/2\,, y_1 + k_{11} \cdot h/2\,, y_2 + k_{12} \cdot h/2)$$

$$k_{32} = f_2(x + h/2\,, y_1 + k_{21} \cdot h/2\,, y_2 + k_{22} \cdot h/2)$$

$$k_{42} = f_2(x + h, y_1 + k_{31}, y_2 + k_{32})$$

где ***h*** – шаг вычисления;

f(**x_i**, **$y_{i,j}$**) – правая часть дифференциального уравнения

ПРИМЕР

Решить систему дифференциальных уравнений первого порядка:

$$\begin{cases} \dfrac{dy_1}{dx} = y_2 \\ \dfrac{dy_2}{dx} = e^{-xy_1} \end{cases}$$

методом Рунге-Кутты на отрезке *[0; 1]* с шагом *h = 0.1*.

Начальные условия: $x_0 = 0$; $y_1(0) = 0$; $y_2(0) = 0$.

РЕШЕНИЕ

Дифференциальное уравнение:

$$\begin{cases} \dfrac{dy_1}{dx} = y_2 \\ \dfrac{dy_2}{dx} = e^{-xy_1} \end{cases}$$

Метод Рунге-Кутты:

$$\begin{cases} y_{(i+1),1} = y_{i,1} + (k_{11} + 2 \cdot k_{21} + 2 \cdot k_{31} + k_{41}) \cdot\ h / 6 \\ y_{(i+1),2} = y_{i,2} + (k_{12} + 2 \cdot k_{22} + 2 \cdot k_{32} + k_{42}) \cdot h / 6 \\ x_{i+1} = x_i + h \end{cases}$$

i	x	k_{11}	k_{21}	k_{31}	k_{41}	$y_{i,1}$	k_{12}	k_{22}	k_{32}	k_{42}	$y_{i,2}$
0	0.0	-	-	-	-	0.0000	-	-	-	-	0.0000
1	0.1	0.0000	0.0500	0.0500	0.1000	0.0050	1.0000	1.0000	0.9999	0.9995	0.1000
2	0.2	0.1000	0.1500	0.1499	0.1998	0.0200	0.9995	0.9985	0.9981	0.9960	0.1998
3	0.3	0.1998	0.2496	0.2494	0.2990	0.0449	0.9960	0.9925	0.9919	0.9866	0.2990
4	0.4	0.2990	0.3483	0.3480	0.3968	0.0797	0.9866	0.9793	0.9784	0.9686	0.3968
5	0.5	0.3968	0.4453	0.4446	0.4923	0.1242	0.9686	0.9562	0.9551	0.9398	0.4924
6	0.6	0.4924	0.5393	0.5384	0.5844	0.1781	0.9398	0.9214	0.9202	0.8987	0.5844
7	0.7	0.5844	0.6293	0.6281	0.6716	0.2409	0.8987	0.8739	0.8727	0.8448	0.6717
8	0.8	0.6717	0.7139	0.7124	0.7529	0.3122	0.8448	0.8139	0.8126	0.7790	0.7529
9	0.9	0.7529	0.7919	0.7901	0.8271	0.3913	0.7790	0.7427	0.7415	0.7032	0.8271
10	1.0	0.8271	0.8623	0.8603	0.8933	0.4774	0.7032	0.6630	0.6619	0.6204	0.8933

РЕАЛИЗАЦИЯ

```
function equations(x: real; y: array of real): array of real;
begin
  result := ArrFill(y.Length, 0.0);
  result[0] := y[1];
  result[1] := exp(-x * y[0])
end;

function RK(f: function(x: real; y: array of real): array of real;
            x0, xf: real; y0: array of real;
            h: real): array of array of real;

  function sum(a: real; arr1, arr2: array of real): array of real;
  begin
    result := ArrFill(arr1.Length, 0.0);
    for var i := 0 to result.High do
      result[i] += arr1[i] + a * arr2[i]
  end;

begin
  var count := Trunc((xf - x0) / h) + 1;
  SetLength(result, count);
  for var i := 0 to result.High do
    SetLength(result[i], y0.Length);

  var x := x0;
  var k1, k2, k3, k4: array of real;
  result[0] := y0;

  for var i := 1 to count-1 do
  begin
    k1 := f(x, result[i-1]);
    k2 := f(x + h / 2, sum(h / 2, result[i-1], k1));
    k3 := f(x + h / 2, sum(h / 2, result[i-1], k2));
    k4 := f(x + h, sum(h, result[i-1], k3));

    for var j := 0 to result[i].High do
      result[i][j] := result[i-1][j] + h / 6
                      * (k1[j] + 2 * k2[j] +
                         2 * k3[j] + k4[j]);


    x += h
  end
end;
```

РЕАЛИЗАЦИЯ


```
begin
  var y0 := Arr(0.0, 0.0);
  var y := RK(equations, 0, 1, y0, 0.1);

  for var i := 0 to y.High do
  begin
    for var j := 0 to y[i].High do
      write(y[i][j]:8:4);
    writeln
  end
end.
```

```
  0.0000  0.0000
  0.0050  0.1000
  0.0200  0.1998
  0.0449  0.2990
  0.0797  0.3968
  0.1242  0.4924
  0.1781  0.5844
  0.2409  0.6717
  0.3122  0.7529
  0.3913  0.8271
  0.4774  0.8933
```

ПРИМЕР

Решите систему дифференциальных уравнений методом Эйлера. Определите погрешность расчетного значения переменной ***y*** для каждого из методов, использовав формулу: $\Delta_i = \frac{|\widetilde{y_i} - y_{ai}|}{y_{ai}} \cdot 100\%$,

где $\widetilde{y_i}$ - расчетное значение, y_{ai} - значение, полученное из аналитического решения.

$$\begin{cases} \dfrac{dy_1}{dx} = -2 \cdot y_1 + 4 \cdot y_2 \\ \dfrac{dy_2}{dx} = -y_1 + 3 \cdot y_2 \end{cases}$$

на отрезке *[0; 1]* с шагом *h = 0.1*.

Начальные условия: $x_0 = 0$; $y_1(0) = 3$; $y_2(0) = 0$.

Аналитическое решение:

$$y_1 = 4 \cdot e^{-x} - e^{2 \cdot x}$$

$$y_2 = e^{-x} - e^{2 \cdot x}$$

РЕАЛИЗАЦИЯ


```
function equations(x: real; y: array of real): array of real;
begin
  result := ArrFill(y.Length, 0.0);
  result[0] := -2 * y[0] + 4 * y[1];
  result[1] := -y[0] + 3 * y[1]
end;


function Eiler(f: function(x: real; y: array of real): array of real;
               x0, xf: real; y0: array of real; h: real): array of array of real;
begin
  var count := Trunc((xf - x0) / h) + 1;
  SetLength(result, count);
  for var i := 0 to result.High do
    SetLength(result[i], y0.Length);

  var x := x0;
  var right_parts := ArrFill(y0.Length, 0.0);
  result[0] := y0;
  for var i := 1 to count-1 do
  begin
    right_parts := f(x, result[i-1]);
    for var j := 0 to result[i].High do
      result[i][j] := result[i-1][j] + h * right_parts[j];
    x += h
  end
end;
```

РЕАЛИЗАЦИЯ

```
function fa(x: real): array of real;
begin
  result := ArrFill(2, 0.0);
  result[0] := 4 * exp(-x) - exp(2 * x);
  result[1] := exp(-x) - exp(2 * x)
end;


begin
  var y0 := Arr(3.0, 0.0);
  var y := Eiler(equations, 0, 1, y0, 0.1);

  for var i := 0 to y.High do
  begin
    for var j := 0 to y[i].High do
      write(y[i][j]:8:4);
    writeln
  end;
  println;
  var x := 0.0;
  for var i := 0 to y.High do
  begin
    var ya := fa(x);
    for var j := 0 to y[i].High do
      write(abs((y[i][j] - ya[j]) / ya[j]) * 100:8:4);
    writeln;
    x += 0.1
  end
end.
```

```
  3.0000  0.0000
  2.4000 -0.3000
  1.8000 -0.6300
  1.1880 -0.9990
  0.5508 -1.4175
 -0.1264 -1.8978
 -0.8602 -2.4545
 -1.6700 -3.1049
 -2.5779 -3.8693
 -3.6101 -4.7724
 -4.7970 -5.8431

  0.0000     NaN
  0.0856  5.2328
  0.9479  6.4024
  4.1051  7.6113
 20.8586  8.8554
 56.7496 10.1300
 23.5272 11.4301
 19.2795 12.7502
 18.3086 14.0852
 18.3858 15.4298
 18.9355 16.7795
```

Задание

Решите системы дифференциальных уравнений методами Эйлера и Рунге-Кутты. Определите погрешность расчетных значений переменных y_i для каждого из методов, используя формулу:

$$\Delta_i = \frac{|\tilde{y}_i - y_{ai}|}{y_{ai}} \cdot 100\%$$

где $\tilde{y}_i$ - расчетное значение, y_{ai} - значение, полученное из аналитического решения.

Система дифф. уравнений	Отрезок, шаг	Начальные условия	Аналитическое решение
1. $\begin{cases} \frac{dy_1}{dx} = -y_1 - 5 \cdot y_2 \\ \frac{dy_2}{dx} = -7 \cdot y_1 - 3 \cdot y_2 \end{cases}$	[0; 0.1]; h = 0.01	$y_1(0) = 13$ $y_2(0) = 11$	$y_1 = 10 \cdot e^{-8 \cdot x} + 3 \cdot e^{4 \cdot x}$ $y_2 = 14 \cdot e^{-8 \cdot x} - 3 \cdot e^{4 \cdot x}$
2. $\begin{cases} \frac{dy_1}{dx} = 3 \cdot y_1 - y_2 \\ \frac{dy_2}{dx} = 4 \cdot y_1 - y_2 \end{cases}$	[0; 0.1]; h = 0.01	$y_1(0) = 5$ $y_2(0) = 8$	$y_1 = (5 + 2 \cdot x) \cdot e^x$ $y_2 = (8 + 4 \cdot x) \cdot e^x$
3. $\begin{cases} \frac{dy_1}{dx} = 2 \cdot y_1 - 5 \cdot y_2 + 3 \\ \frac{dy_2}{dx} = 5 \cdot y_1 - 6 \cdot y_2 + 1 \end{cases}$	[2; 3]; h = 0.1	$y_1(0) = 1.0879$ $y_2(0) = 1.0550$	$y_1 = 5 \cdot e^{-2 \cdot x} \cdot \cos(3 \cdot x) + 1$ $y_2 = e^{-2 \cdot x} \cdot (4 \cdot \cos(3 \cdot x) + 3 \cdot \sin(3 \cdot x)) + 1$
4. $\begin{cases} \frac{dy_1}{dx} = y_1 - 2 \cdot y_2 \\ \frac{dy_2}{dx} = y_1 - y_2 - 2 \end{cases}$	[0; 1]; h = 0.1	$y_1(0) = 1$ $y_2(0) = -2$	$y_1 = -3 \cdot \cos(x) + 5 \cdot \sin(x) + 4$ $y_2 = -4 \cdot \cos(x) + \sin(x) + 2$

ТОМСКИЙ ПОЛИТЕХНИЧЕСКИЙ УНИВЕРСИТЕТ

КОНТАКТНАЯ ИНФОРМАЦИЯ

ЧУЗЛОВ ВЯЧЕСЛАВ АЛЕКСЕЕВИЧ

к.т.н., доцент ОХИ ИШПР

Учебный корпус №2, ауд. 136

+7-962-782-66-15

chuva@tpu.ru