

Углубленный курс информатики

Численные методы решения обыкновенных
дифференциальных уравнений



Чузлов Вячеслав Алексеевич

к.т.н., доцент ОХИ ИШПР

МЕТОД ЭЙЛЕРА

- Пусть дано дифференциальное уравнение:

$$\frac{dy}{dx} = f(x, y)$$

с начальным условием:

$$y|_{x=x_0} = y_0$$

$$y = y(x) - ?$$

Формула Эйлера:

$$\begin{cases} y_{i+1} = y_i + h \cdot f(x_i, y_i) \\ x_{i+1} = x_i + h \end{cases}$$

где h – шаг вычисления;

$f(x_i, y_i)$ – правая часть дифференциального уравнения

ПРИМЕР

Решить дифференциальное уравнение первого порядка:

$$\frac{dy}{dx} = \frac{y}{(\cos(x))^2}$$

методом Эйлера на отрезке $[0; 1]$ с шагом $h = 0.1$.

Начальные условия: $x_0 = 0$; $y_0 = 2.7183$.

РЕШЕНИЕ

Дифференциальное уравнение:

$$\frac{dy}{dx} = \frac{y}{(\cos(x))^2}$$

Метод Эйлера:

$$\begin{cases} y_{i+1} = y_i + h \cdot f(x_i, y_i) \\ x_{i+1} = x_i + h \end{cases}$$

$$1. x_0 = 0; y_0 = 2.7183$$

$$2. y_1 = y_0 + h \cdot \frac{y_0}{(\cos(x_0))^2} = 2.7183 + 0.1 \cdot \frac{2.7183}{(\cos(0))^2} = 2.9901$$

$$3. y_2 = y_1 + h \cdot \frac{y_1}{(\cos(x_1))^2} = 2.9901 + 0.1 \cdot \frac{2.9901}{(\cos(0.1))^2} = 3.2922$$

$$4. y_3 = y_2 + h \cdot \frac{y_2}{(\cos(x_2))^2} = 3.2922 + 0.1 \cdot \frac{3.2922}{(\cos(0.2))^2} = 3.6349$$

.....

$$11. y_{10} = y_9 + h \cdot \frac{y_9}{(\cos(x_9))^2} = 8.2497 + 0.1 \cdot \frac{8.2497}{(\cos(0.9))^2} = 10.3847$$

РЕАЛИЗАЦИЯ

```
function f(x, y: real): real;
begin
    result := y / sqr(cos(x))
end;

function Euler(f: function(x, y: real): real;
               x0, xf, y0, h: real): array of real;
begin
    var count := Trunc((xf - x0) / h)+1;
    result := ArrFill(count, 0.0);

    result[0] := y0;
    var x := x0;
    for var i := 1 to count-1 do
    begin
        result[i] := result[i-1] + h * f(x, result[i-1]);
        x += h
    end;
end;

begin
    var y := Euler(f, 0, 1, 2.7183, 0.1);
    y.PrintLines
end.
```

```
2.7183
2.99013
3.29215317775193
3.63489640017217
4.03316796957362
4.50857929992964
5.09399416073368
5.84181449558991
6.84044330855103
8.24968103879327
10.3846974466126
```

МЕТОД РУНГЕ-КУТТЫ

- Пусть дано дифференциальное уравнение:

$$\frac{dy}{dx} = f(x, y)$$

с начальным условием:

$$y|_{x=x_0} = y_0$$

$$y = y(x) - ?$$

Формула Рунге-Кутты:

$$\begin{cases} y_{i+1} = y_i + \frac{h}{6} \cdot (k_1 + 2 \cdot k_2 + 2 \cdot k_3 + k_4) \\ x_{i+1} = x_i + h \end{cases}$$

$$k_1 = f(x_i, y_i)$$

$$k_2 = f\left(x_i + \frac{h}{2}, y_i + \frac{h}{2} \cdot k_1\right)$$

$$k_3 = f\left(x_i + \frac{h}{2}, y_i + \frac{h}{2} \cdot k_2\right)$$

$$k_4 = f(x_i + h, y_i + h \cdot k_3)$$

где h – шаг вычисления;

$f(x_i, y_i)$ – правая часть дифференциального уравнения

ПРИМЕР

Решить дифференциальное уравнение первого порядка:

$$\frac{dy}{dx} = \frac{y}{(\cos(x))^2}$$

методом Эйлера на отрезке $[0; 1]$ с шагом $h = 0.1$.

Начальные условия: $x_0 = 0$; $y_0 = 2.7183$.

РЕШЕНИЕ

Дифференциальное уравнение:

$$\frac{dy}{dx} = \frac{y}{(\cos(x))^2}$$

Метод Рунге-Кутты:

$$\begin{cases} y_{i+1} = y_i + \frac{h}{6} \cdot (k_1 + 2 \cdot k_2 + 2 \cdot k_3 + k_4) \\ x_{i+1} = x_i + h \end{cases}$$

i	x_i	$k_1 = \frac{y_{i-1}}{(\cos(x_{i-1}))^2}$	$k_2 = \frac{y_{i-1} + \frac{h}{2} \cdot k_1}{\left(\cos\left(x_{i-1} + \frac{h}{2}\right)\right)^2}$	$k_3 = \frac{y_{i-1} + \frac{h}{2} \cdot k_2}{\left(\cos\left(x_{i-1} + \frac{h}{2}\right)\right)^2}$	$k_4 = \frac{y_{i-1} + h \cdot k_3}{(\cos(x_{i-1} + h))^2}$	y_i
0	0.0	-	-	-	-	2.7183
1	0.1	2.7183	2.8614	2.8685	3.0354	3.0052
2	0.2	3.0354	3.2291	3.2390	3.4659	3.3291
3	0.3	3.4659	3.7308	3.7449	4.0580	3.7037
4	0.4	4.0581	4.4272	4.4481	4.8901	4.1487
5	0.5	4.8903	5.4184	5.4509	6.0947	4.6941
6	0.6	6.0951	6.8779	6.9318	7.9088	5.3878
7	0.7	7.9096	9.1256	9.2215	10.7866	6.3110
8	0.8	10.7883	12.7957	12.9832	15.6764	7.6114
9	0.9	15.6806	19.2742	19.6867	24.7932	9.5846
10	1.0	24.8050	31.9927	33.0548	44.1554	12.9022

РЕАЛИЗАЦИЯ

```
function f(x, y: real): real;
begin
    result := y / sqr(cos(x))
end;

function RK(f: function(x, y: real): real;
            x0, xf, y0, h: real): array of real;
begin
    var count := Trunc((xf - x0) / h)+1;
    var (k1, k2, k3, k4) := (0.0, 0.0, 0.0, 0.0);
    result := ArrFill(count, 0.0);


    result[0] := y0;
    var x := x0;
    for var i := 1 to count-1 do
    begin
        k1 := f(x, result[i-1]);
        k2 := f(x + h / 2, result[i-1] + h / 2 * k1);
        k3 := f(x + h / 2, result[i-1] + h / 2 * k2);
        k4 := f(x + h, result[i-1] + h * k3);
        result[i] := result[i-1] + h / 6 * (k1 + 2 * k2 + 2 * k3 + k4);
        x += h
    end;
end;

begin
    var y := RK(f, 0, 1, 2.7183, 0.1);
    y.PrintLines
end.
```

```
2.7183
3.00519163706974
3.32914889838646
3.7037391311616
4.14871996305487
4.69411269398281
5.38783371559872
6.31100442874922
7.61137986737866
9.58463957359314
12.9022298485301
```

АНАЛИТИЧЕСКОЕ РЕШЕНИЕ

i	x_i	$y_i = e^{tg(x)+1}$
0	0.0	2.7183
1	0.1	3.0052
2	0.2	3.3291
3	0.3	3.7037
4	0.4	4.1487
5	0.5	4.6941
6	0.6	5.3878
7	0.7	6.3110
8	0.8	7.6113
9	0.9	9.5846
10	1.0	12.9023



Задание

Решите дифференциальные уравнения методами Эйлера и Рунге-Кутты. Определите погрешность расчетных значений переменной y для каждого из методов, используя формулу:


$$\Delta = \frac{|\tilde{y} - y_a|}{y_a} \cdot 100\% \quad \text{где } \tilde{y} - \text{расчетное значение, } y_a - \text{значение, полученное из аналитического решения.}$$


Дифференциальное уравнение	Отрезок, шаг	Начальные условия	Аналитическое решение
1. $\frac{dy}{dx} = -(2 \cdot y + 1) \cdot \operatorname{ctg}(x)$	[1; 2]; h = 0.1	$x_0 = 1$; $y_0 = 3.0307$	$y = \frac{5}{2 \cdot (\sin(x))^2} - \frac{1}{2}$
2. $\frac{dy}{dx} = -\frac{y \cdot \ln(y)}{x}$	[1; 2]; h = 0.2	$x_0 = 1$; $y_0 = e$	$y = e^{1/x}$
3. $\frac{dy}{dx} = \frac{2 \cdot x}{e^{(y-x^2)}}$	[0; 2]; h = 0.2	$x_0 = 0$; $y_0 = \ln(2)$	$y = \ln(e^{x^2} + 1)$
4. $\frac{dy}{dx} = \frac{y-3}{3 \cdot x^2 + x}$	[2; 3]; h = 0.2	$x_0 = 2$; $y_0 = 4$	$y = \frac{3.5 \cdot x}{3 \cdot x + 1} + 3$


КОНТАКТНАЯ ИНФОРМАЦИЯ

ЧУЗЛОВ ВЯЧЕСЛАВ АЛЕКСЕЕВИЧ

к.т.н., доцент ОХИ ИШПР

 Учебный корпус №2, ауд. 136

 +7-962-782-66-15

 chuva@tpu.ru