

# Углубленный курс информатики

Аппроксимация экспериментальных данных



Чузлов Вячеслав Алексеевич

к.т.н.. доцент ОХИ ИШПР

# ПОСТАНОВКА ЗАДАЧИ АППРОКСИМАЦИИ

Пусть задана функциональная зависимость  $y = f(x)$ , полученная в эксперименте:

<b>X</b>	$x_1$	$x_2$	$x_3$	...	$x_n$
<b>Y</b>	$y_1$	$y_2$	$y_3$	...	$y_n$




$n$  – количество известных экспериментальных значений. Если аналитическое выражение функции  $f(x)$  неизвестно или является сложным, то возникает практически важная задача: найти такую эмпирическую формулу:

$$\tilde{y} = \tilde{f}(x)$$

значения которой при  $x = x_i$  были близки к экспериментальным данным  $y_i$  ( $i = 1, 2, \dots, n$ ).

# ПОСТАНОВКА ЗАДАЧИ АППРОКСИМАЦИИ

Графическое сравнение аппроксимирующих функций:



- В данном случае экспоненциальная аппроксимация точнее описывает экспериментальные данные.
- Для других экспериментальных данных возможны другие функции, которые будут точнее их описывать.

# ЛИНЕЙНАЯ АППРОКСИМАЦИЯ

$$\tilde{y} = a_0 + a_1 x$$

$a_0$  ,  $a_1$  – коэффициенты линейной аппроксимирующей функции

- Для линейной аппроксимирующей функции коэффициенты  $a_0$  и  $a_1$  определяются по методу наименьших квадратов:

$$F = \sum_{i=1}^n [y_i - (a_0 + a_1 x_i)]^2 \rightarrow \min$$

где  $y_i$  - табличные значения функции;  $a_0 + a_1 x_i$  - линейная аппроксимирующая функция

# МЕТОД НАИМЕНЬШИХ КВАДРАТОВ

$$\begin{cases} \frac{\partial F}{\partial a_0} = -2 \cdot \sum_{i=1}^n (y_i - a_0 - a_1 \cdot x_i) \cdot 1 = 0 \\ \frac{\partial F}{\partial a_1} = -2 \cdot \sum_{i=1}^n (y_i - a_0 - a_1 \cdot x_i) \cdot x_i = 0 \end{cases} \quad \begin{cases} \sum_{i=1}^n y_i - a_0 \cdot n - a_1 \cdot \sum_{i=1}^n x_i = 0 \\ \sum_{i=1}^n (y_i \cdot x_i) - a_0 \cdot \sum_{i=1}^n x_i - a_1 \cdot \sum_{i=1}^n x_i^2 = 0 \end{cases}$$

$$\begin{cases} a_0 \cdot n + a_1 \cdot \sum_{i=1}^n x_i = \sum_{i=1}^n y_i \\ a_0 \cdot \sum_{i=1}^n x_i + a_1 \cdot \sum_{i=1}^n x_i^2 = \sum_{i=1}^n (y_i \cdot x_i) \end{cases}$$

# МЕТОД НАИМЕНЬШИХ КВАДРАТОВ

$$a_0 = \frac{\begin{vmatrix} \sum_{i=1}^n y_i & \sum_{i=1}^n x_i \\ \sum_{i=1}^n (x_i \cdot y_i) & \sum_{i=1}^n x_i^2 \end{vmatrix}}{\begin{vmatrix} n & \sum_{i=1}^n x_i \\ \sum_{i=1}^n x_i & \sum_{i=1}^n x_i^2 \end{vmatrix}} = \frac{\sum_{i=1}^n y_i \cdot \sum_{i=1}^n x_i^2 - \sum_{i=1}^n x_i \cdot \sum_{i=1}^n (x_i \cdot y_i)}{n \cdot \sum_{i=1}^n x_i^2 - \left( \sum_{i=1}^n x_i \right)^2}$$

$$a_0 = \frac{\begin{vmatrix} n & \sum_{i=1}^n y_i \\ \sum_{i=1}^n x_i & \sum_{i=1}^n (x_i \cdot y_i) \end{vmatrix}}{\begin{vmatrix} n & \sum_{i=1}^n x_i \\ \sum_{i=1}^n x_i & \sum_{i=1}^n x_i^2 \end{vmatrix}} = \frac{n \cdot \sum_{i=1}^n (x_i \cdot y_i) - \sum_{i=1}^n x_i \cdot \sum_{i=1}^n y_i}{n \cdot \sum_{i=1}^n x_i^2 - \left( \sum_{i=1}^n x_i \right)^2}$$

$$S_1 = \sum_{i=1}^n x_i \quad S_2 = \sum_{i=1}^n y_i$$

$$S_3 = \sum_{i=1}^n x_i^2 \quad S_4 = \sum_{i=1}^n x_i \cdot y_i$$

**Коэффициенты аппроксимации:**

$$a_0 = \frac{S_2 \cdot S_3 - S_1 \cdot S_4}{n \cdot S_3 - S_1^2}$$

$$a_1 = \frac{n \cdot S_4 - S_1 \cdot S_2}{n \cdot S_3 - S_1^2}$$

## ПРИМЕР

Составить программу для аппроксимации экспериментальных данных:

$X$	300	400	500	600	700	800
$Y$	6.97	7.01	7.12	7.28	7.45	7.62

Вычислить значение  $Y$  в точке  $X = 750$ .

# Реализация

```
function LineFit(x, y: array of real): array of real;
begin
    var (s1, s2, s3, s4) := (0.0, 0.0, 0.0, 0.0);
    result := ArrFill(2, 0.0);

    for var i := 0 to x.High do
    begin
        s1 += x[i];
        s2 += y[i];
        s3 += sqr(x[i]);
        s4 += x[i] * y[i]
    end;

    var n := x.Length;
    result[0] := (s2 * s3 - s1 * s4) / (n * s3 - sqr(s1));
    result[1] := (n * s4 - s1 * s2) / (n * s3 - sqr(s1))
end;

function line_func(a: array of real; x: real): real;
begin
    result := a[0] + a[1] * x
end;
```

```
begin
    var x: array of real = (300, 400, 500,
                             600, 700, 800);

    var y: array of real = (6.97, 7.01, 7.12,
                             7.28, 7.45, 7.62);

    var a := LineFit(x, y);
    var y_ := line_func(a, 750);
    println('$'При x = 750 y = {y_}')
end.
```

При x = 750 y = 7.51195238095238




## ЭКСПОНЕНЦИАЛЬНАЯ АППРОКСИМАЦИЯ

$$P(x) = A * e^{k \cdot x}$$

$$\ln(P(x)) = \ln(A) + k \cdot \ln(e) \cdot x$$

Сделаем замену:  $\ln(P(x)) = y$ ;  $\ln(A) = a_0$ ;  $k = a_1$ :



$$y = a_0 + a_1 \cdot x$$

# Реализация

```
function ExpFit(x, y: array of real): array of real;
begin
    var (s1, s2, s3, s4) := (0.0, 0.0, 0.0, 0.0);
    result := ArrFill(2, 0.0);

    for var i := 0 to x.High do
    begin
        s1 += x[i];
        s2 += ln(y[i]);
        s3 += sqr(x[i]);
        s4 += x[i] * ln(y[i])
    end;

    var n := x.Length;
    result[0] := (s2 * s3 - s1 * s4) / (n * s3 - sqr(s1));
    result[1] := (n * s4 - s1 * s2) / (n * s3 - sqr(s1))
end;

function exp_func(a: array of real; x: real): real;
begin
    result := exp(a[0]) * exp(a[1] * x)
end;
```

```
begin
    var x: array of real = (300, 400, 500,
                           600, 700, 800);

    var y: array of real = (6.97, 7.01, 7.12,
                           7.28, 7.45, 7.62);

    var a := ExpFit(x, y);
    var y_ := exp_func(a, 750);
    println('$'При x = 750 y = {y_}')
end.
```


При x = 750 y = 7.51206780537444

## СТЕПЕННАЯ АППРОКСИМАЦИЯ

$$P(x) = A * x^k$$

$$\ln(P(x)) = \ln(A) + k \cdot \ln(x)$$

Сделаем замену:  $\ln(P(x)) = y$ ;  $\ln(A) = a_0$ ;  $k = a_1$ :



$$y = a_0 + a_1 \cdot \ln(x)$$

# Реализация

```
function PowFit(x, y: array of real): array of real;
begin
    var (s1, s2, s3, s4) := (0.0, 0.0, 0.0, 0.0);
    result := ArrFill(2, 0.0);

    for var i := 0 to x.High do
    begin
        s1 += ln(x[i]);
        s2 += ln(y[i]);
        s3 += sqr(ln(x[i]));
        s4 += ln(x[i]) * ln(y[i])
    end;

    var n := x.Length;
    result[0] := (s2 * s3 - s1 * s4) / (n * s3 - sqr(s1));
    result[1] := (n * s4 - s1 * s2) / (n * s3 - sqr(s1))
end;

function pow_func(a: array of real; x: real): real;
begin
    result := exp(a[0]) * x ** a[1]
end;
```

```
begin
    var x: array of real = (300, 400, 500,
                           600, 700, 800);

    var y: array of real = (6.97, 7.01, 7.12,
                           7.28, 7.45, 7.62);

    var a := PowFit(x, y);
    var y_ := pow_func(a, 750);
    println('$'При x = 750 y = {y_}')
end.
```

При x = 750 y = 7.48357339816782

# Задание

*Дана зависимость теплоемкости и энтальпии от температуры:*

$T, K$	<b>Этилбензол <math>C_8H_{10}</math></b>	
	$C_p, ^\circ, Дж/(моль \cdot K)$	$\Delta H, кДж/моль$
300	129.20	29.62
400	170.54	21.88
500	206.48	15.52
600	236.14	10.38
700	260.58	6.40
800	280.96	3.35
900	298.19	1.13
1000	312.84	0.21

1. С использованием линейной, экспоненциальной и степенной аппроксимации определить значения теплоемкости и энтальпии при изменении  $T$  в интервале от 300 до 1000 с шагом 50.
2. Построить графики в Excel по табличным данным и результатам аппроксимации. Выбрать аппроксимирующую функцию, которая наиболее точно описывает табличные данные.





# КОНТАКТНАЯ ИНФОРМАЦИЯ


---

**ЧУЗЛОВ ВЯЧЕСЛАВ АЛЕКСЕЕВИЧ**

к.т.н., доцент ОХИ ИШПР

 Учебный корпус №2, ауд. 136

 +7-962-782-66-15

 [chuva@tpu.ru](mailto:chuva@tpu.ru)