ТОМСКИЙ
ПОЛИТЕХНИЧЕСКИЙ
УНИВЕРСИТЕТ

tpu.ru

Углубленный курс информатики

Численные методы решения определенных интегралов

Чузлов Вячеслав Алексеевич

к.т.н., доцент ОХИ ИШПР

ПОСТАНОВКА ЗАДАЧИ

Вычислить определенный интеграл:

$$I = \int_a^b f(x)dx$$

при условии, что *a* и *b* конечны и $f(x)$ является непрерывной функцией *x* на всем интервале $x \in [a, b]$.

Во многих случаях, когда подынтегральная функция задана в аналитическом виде, интеграл от этой функции в пределах от *a* до *b* может быть вычислен по формуле Ньютона–Лейбница:

$$\int_a^b f(x)dx = F(x)\Big|_b^a = F(b) - F(a)$$



Однако этой формулой часто нельзя воспользоваться по следующим причинам:

- первообразная функция F(x) слишком сложна и ее нельзя выразить в элементарных функциях;
- функция f(x) задана в виде таблицы, что особенно часто встречается в задачах химической технологии при обработке экспериментальных данных.

Задача численного интегрирования состоит в нахождении приближенного значения интеграла по заданным или вычисленным значениям.

МЕТОД ПРЯМОУГОЛЬНИКОВ

Метод заключается в приближенном вычислении площади под графиком суммированием площадей конечного числа прямоугольников, ширина которых будет определяться расстоянием между соответствующими соседними узлами интегрирования, высота – значением подынтегральной функции в этих узлах.

$$\int_a^b f(x)dx \approx h \cdot \sum_{i=0}^{n-1} f\big(a + h \cdot (i + 0.5)\big)$$

где $h = \frac{(b-a)}{n}$, n – число интервалов разбиения



ПРИМЕР

Вычислите интеграл методом прямоугольников:

$$I = \int_0^1 \frac{dx}{1 + x^2}$$

Интервал интегрирования разбейте на 10 равных частей (n = 10).

РЕАЛИЗАЦИЯ

```
function f(x: real): real;
begin
  result := 1 / (1 + x ** 2)
end;


function RectMethod(f: function(x: real): real;
                    x: array of real; n: integer): real;
begin
  var h := (x[1] - x[0]) / n;
  result := 0.0;
  for var i := 0 to n-1 do
    result += f(x[0] + h * (i + 0.5));
  result *= h
end;


begin
  var x := Arr(0.0, 1.0);
  var int := RectMethod(f, x, 10);
  int.Println
end.
```

```
0.785606496250275
```

МЕТОД ТРАПЕЦИЙ

Метод численного интегрирования функции одной переменной, заключающийся в замене на каждом элементарном отрезке подынтегральной функции на многочлен первой степени, то есть линейную функцию. Площадь под графиком функции аппроксимируется прямоугольными трапециями.

$$\int_a^b f(x)dx \approx h \cdot \sum_{i=0}^{n-1} \frac{f(x_i) + f(x_{i+1})}{2}$$

где $h = \frac{(b-a)}{n}$, n – число интервалов разбиения



ПРИМЕР

Вычислите интеграл методом трапеций:

$$I = \int_0^1 \frac{dx}{1 + x^2}$$

Интервал интегрирования разбейте на 10 равных частей (n = 10).

РЕАЛИЗАЦИЯ


```
function f(x: real): real;
begin
  result := 1 / (1 + x ** 2)
end;


function TrapMethod(f: function(x:real):real;
                    x: array of real; n: integer): real;
begin
  var h := (x[1] - x[0]) / n;
  result := 0.0;

  var x_ := new real[n+1];
  (x_[0], x_[^1]) := (x[0], x[1]);
  for var i := 1 to x_.High-1 do
    x_[i] := x_[i-1] + h;

  for var i := 0 to x_.High-1 do
    result += (f(x_[i]) + f(x_[i+1])) / 2;
  result *= h
end;


begin
  var x := Arr(0.0, 1.0);
  var int := TrapMethod(f, x, 10);
  int.Println;
end.
```

```
0.78498149722679
```

ФОРМУЛА СИМПСОНА (МЕТОД ПАРАБОЛ)

Суть метода заключается в приближении подынтегральной функции на отрезке $[a, b]$ интерполяционным полиномом второй степени, т.е. приближение графика функции на отрезке параболой.

$$\int_a^b f(x)dx \approx \frac{h}{3} \cdot \left(f(x_0) + 2 \cdot \sum_{i=1}^{\frac{N}{2}-1} f(x_{2i}) + 4 \cdot \sum_{i=1}^{\frac{N}{2}} f(x_{2i-1}) + f(x_N) \right)$$

где $h = \frac{(b-a)}{n}$, N = 2n – число интервалов разбиения



ПРИМЕР

Вычислите интеграл, используя формулу Симпсона:

$$I = \int_0^1 \frac{dx}{1 + x^2}$$

Интервал интегрирования разбейте на 10 равных частей (N = 10).

РЕАЛИЗАЦИЯ


```
function f(x: real): real;
begin
  result := 1 / (1 + x ** 2)
end;


function Simpson(f: function(x:real):real;
                 x: array of real; n: integer): real;
begin
  var h := (x[1] - x[0]) / n;
  var (s1, s2) := (0.0, 0.0);

  for var i := 1 to n-1 do
    if i mod 2 = 0 then
      s1 += f(x[0] + i * h)
    else
      s2 += f(x[0] + i * h);

  result := h / 3 * (f(x[0]) + 2 * s1 + 4 * s2 + f(x[1]))
end;


begin
  var x := Arr(0.0, 1.0);
  var int := Simpson(f, x, 10);
  int.Println;
end.
```

```
0.785398153484804
```

Задание 1

Составьте программу для вычисления определенных интегралов по формулам прямоугольников, трапеций и Симпсона:

#	Интеграл	Число интервалов разбиения
1.	$\int_{0}^{\sqrt{3}} \frac{x\,dx}{\sqrt{x^4+16}}$	$n = 10$
2.	$\int_{\frac{\pi}{2}}^{\pi} \frac{\sin(x)\,dx}{\left(cos(x)\right)^2+1}$	$n = 20$
3.	$\int_{1}^{2} \frac{e^{\frac{1}{x}}\,dx}{x^2}$	$n = 10$

Задание 2

Используя формулы прямоугольников, трапеций и Симпсона, вычислите значение энтропии воды при ее нагревании от 400 до 500 К по формуле:

$$\Delta S = \eta \cdot \int_{400}^{500} \frac{C_v \cdot dT}{T}$$

Количество молей $\eta = 3$,
значение теплоемкости $C_v = 35.0\ \text{Дж}/(\text{моль} \cdot \text{К})$.

Интервал по температурам разбейте на 10 равных частей.

ТОМСКИЙ ПОЛИТЕХНИЧЕСКИЙ УНИВЕРСИТЕТ

КОНТАКТНАЯ ИНФОРМАЦИЯ

ЧУЗЛОВ ВЯЧЕСЛАВ АЛЕКСЕЕВИЧ

к.т.н., доцент ОХИ ИШПР

Учебный корпус №2, ауд. 136

+7-962-782-66-15

chuva@tpu.ru